2 - 11 Second shifting theorem, unit step function
Sketch or graph the given function, which is assumed to be zero outside the given interval. Represent it, using unit step functions. Find its transform.

3. t - 2 (t > 2)

```
ClearAll["Global`*"]
```

```
e1 = LaplaceTransform[Piecewise[{{t - 2, t > 2}, {0, t < 2}}], t, s]
```

$$\frac{e^{-2 s}}{s^2}$$

The above answer matches the text.

5. $e^t \quad \left(0 < t < \frac{\pi}{2}\right)$

```
ClearAll["Global`*"]
```

$$\text{e1} = \text{LaplaceTransform}\left[\text{Piecewise}\left[\left\{\left\{e^t,\ 0 < t < \frac{\pi}{2}\right\},\ \left\{0,\ t < 0\ \&\&\ t > \frac{\pi}{2}\right\}\right\}\right],\ t,\ s\right]$$

$$\frac{e^{-\frac{1}{2}\pi(-1+s)}\left(-1 + e^{\frac{1}{2}\pi(-1+s)}\right)}{-1 + s}$$

```
e2 = Expand[e1]
```

$$\frac{1}{-1 + s} - \frac{e^{-\frac{1}{2}\pi(-1+s)}}{-1 + s}$$

The above answer matches the text.

7. $e^{-\pi t}$ (2 < t < 4)

```
ClearAll["Global`*"]
```

$$\text{e1} = \text{LaplaceTransform}\left[\text{Piecewise}\left[\left\{\left\{e^{-\pi t},\ 2 < t < 4\right\},\ \{0,\ t < 2\ \&\&\ t > 4\}\right\}\right],\ t,\ s\right]$$

$$\frac{e^{-4\pi - 4 s}\left(-1 + e^{2\pi + 2 s}\right)}{\pi + s}$$

The above answer matches the text.

9. $t^2 \quad \left(t > \frac{3}{2}\right)$

```
ClearAll["Global`*"]
```

$$\text{e1} = \text{LaplaceTransform}\left[\text{Piecewise}\left[\left\{\left\{t^2, t > \frac{3}{2}\right\}, \left\{0, t < \frac{3}{2}\right\}\right\}\right], t, s\right]$$

$$\frac{e^{-3 s/2} \left(8 + 12 s + 9 s^2\right)}{4 s^3}$$

The above answer matches the text.

11. $\text{Sin}[t] \quad \left(\frac{\pi}{2} < t < \pi\right)$

```
ClearAll["Global`*"]
```

$$\text{e1} = \text{LaplaceTransform}\left[\text{Piecewise}\left[\left\{\left\{\text{Sin}[t], \frac{\pi}{2} < t < \pi\right\}, \left\{0, t < \frac{\pi}{2} \ \&\& \ t > \pi\right\}\right\}\right], t, s\right]$$

$$\frac{e^{-\pi s} \left(1 + e^{\frac{\pi s}{2}} s\right)}{1 + s^2}$$

The above answer matches the text.

12 - 17 Inverse transforms by the 2nd shifting theorem
Find and sketch or graph f(t) if $\mathcal{L}$(f) equals

13. $\frac{6 \left(1 - e^{-\pi s}\right)}{\left(s^2 + 9\right)}$

```
ClearAll["Global`*"]
```

$$\text{e1} = \text{InverseLaplaceTransform}\left[\frac{6 \left(1 - e^{-\pi s}\right)}{\left(s^2 + 9\right)}, s, t\right]$$

$$2 \left(1 + \text{HeavisideTheta}[-\pi + t]\right) \text{Sin}[3 t]$$

The above answer is not exactly like the text. That answer refers to $u(t - \pi)$ instead of **Heaviside**, but the two plots below make me think they are equivalent.

```
Plot[UnitStep[-π + t], {t, -6, 6}, ImageSize → 150]
```



```
Plot[HeavisideTheta[-π + t], {t, -6, 6}, ImageSize → 150]
```



```
e2 = Plot[e1, {t, 0, 4 π}, PlotRange → Automatic,
  PlotStyle → Thickness[0.003], ImageSize → 250]
```



The above plot looks just like the one in the s.m..

15. $e^{\frac{-3 s}{s^4}}$

```
ClearAll["Global`*"]
```

e1 = InverseLaplaceTransform$\left[\frac{e^{-3 s}}{s^4}, s, t\right]$

$\frac{1}{6}(-3+t)^3$ HeavisideTheta[-3 + t]

The above answer matches the text.

```
e2 = Plot[e1, {t, 0, 4 π}, PlotRange → Automatic,
  PlotStyle → Thickness[0.003], ImageSize → 250]
```



17. $\frac{\left(1+e^{-2\pi(s+1)}\right)(s+1)}{\left((s+1)^2+1\right)}$

```
ClearAll["Global`*"]
```

```
e1 = InverseLaplaceTransform[((1 + e^(-2 π (s+1))) (s + 1))/((s + 1)^2 + 1), s, t]
```

$\frac{1}{2} e^{(-1-i)t}\left(1+e^{2it}\right)(1+\text{HeavisideTheta}[-2\pi+t])$

```
e2 = FullSimplify[e1]
```

$e^{-t}\,\text{Cos}[t]\,(1+\text{HeavisideTheta}[-2\pi+t])$

The above answer agrees with the text answer when $t < 2\pi$. Though not stated in the problem, the text answer clarifies that interval is all the text answer is willing to consider.

```
e3 = Plot[e1, {t, 0, 3 π}, PlotRange → {-.1, 0.05},
   PlotStyle → {Yellow, Thickness[0.003]}, ImageSize → 350];
```

```
e4 = Plot[e^-t Cos[t], {t, 0, 3 π}, PlotRange → Automatic,
   PlotStyle → {Red, Thickness[0.03]}, ImageSize → 250];
Show[
 e4,
 e3]
```



```
e5 = Plot[e1, {t, 5, 2 π + 0.2}, PlotRange → Automatic,
   PlotStyle → {Yellow, Thickness[0.003]}, ImageSize → 350];
```

```
e6 = Plot[ⅇ^-t Cos[t], {t, 5, 2 π + 0.2}, PlotRange → {-.05, 0.1},
   PlotStyle → {Red, Thickness[0.03]}, ImageSize → 250];
```

```
Show[e6, e5]
```



18 - 27 IVPs, some with discontinuous input
Using the Laplace transform and showing the details, solve

19. $y'' + 6y' + 8y = e^{-3t} - e^{5t}$, $y[0] = 0$, $y'[0] = 0$

```
ClearAll["Global`*"]
```

```
e1 = LaplaceTransform[y''[t] + 6 y'[t] + 8 y[t] == ⅇ^-3 t - ⅇ^-5 t, t, s]
```

$$8\,\text{LaplaceTransform}[y[t], t, s] + s^2\,\text{LaplaceTransform}[y[t], t, s] + 6\,(s\,\text{LaplaceTransform}[y[t], t, s] - y[0]) - s\,y[0] - y'[0] == \frac{1}{3+s} - \frac{1}{5+s}$$

```
e2 = e1 /. {y[0] → 0, y'[0] → 0, LaplaceTransform[y[t], t, s] → yoft}
```

$$8\,\text{yoft} + 6\,s\,\text{yoft} + s^2\,\text{yoft} == \frac{1}{3+s} - \frac{1}{5+s}$$

```
e3 = Solve[e2, yoft]
```

$$\left\{\left\{\text{yoft} \to \frac{2}{(3+s)\,(5+s)\,\left(8+6\,s+s^2\right)}\right\}\right\}$$

```
e4 = e3[[1, 1, 2]]
```

$$\frac{2}{(3+s)\,(5+s)\,\left(8+6\,s+s^2\right)}$$

```
e5 = InverseLaplaceTransform[e4, s, t]
```

$$\frac{1}{3}\,e^{-5t}\left(-1+e^{t}\right)^3$$

```
y[t_] = 1/3 e^(-5 t) (-1 + e^t)^3
```

$$\frac{1}{3} e^{-5t} \left(-1 + e^t\right)^3$$

```
y[0]
```

0

```
y'[0]
```

0

The above answer in the green cell matches the text.

21. y" + 9 y = 8 Sin[t] if 0 < t < π and 0 if t > π; y[0] = 0, y'[0] = 4

```
ClearAll["Global`*"]
```

Note that there is no shifting here; only a piecewise function definition.

```
e1 = LaplaceTransform[y''[t] + 9 y[t] == 8 Sin[t] (1 - UnitStep[t - π]), t, s]
```

$$9\,\text{LaplaceTransform}[y[t], t, s] + s^2\,\text{LaplaceTransform}[y[t], t, s] - s\,y[0] - y'[0] == 8\left(\frac{1}{1+s^2} + \frac{e^{-\pi s}}{1+s^2}\right)$$

```
e2 = e1 /. {y[0] → 0, y'[0] → 4, LaplaceTransform[y[t], t, s] → bigY}
```

$$-4 + 9\,\text{bigY} + \text{bigY}\,s^2 == 8\left(\frac{1}{1+s^2} + \frac{e^{-\pi s}}{1+s^2}\right)$$

```
e3 = Solve[e2, bigY]
```

$$\left\{\left\{\text{bigY} \to \frac{4 e^{-\pi s}\left(2 + 3 e^{\pi s} + e^{\pi s} s^2\right)}{\left(1+s^2\right)\left(9+s^2\right)}\right\}\right\}$$

```
e4 = e3[[1, 1, 2]]
```

$$\frac{4 e^{-\pi s}\left(2 + 3 e^{\pi s} + e^{\pi s} s^2\right)}{\left(1+s^2\right)\left(9+s^2\right)}$$

```
e5 = InverseLaplaceTransform[e4, s, t]
```

$$4\left(\text{Cos}[t]^2\,\text{Sin}[t] - \frac{1}{3}\,\text{HeavisideTheta}[-\pi + t]\,\text{Sin}[t]^3\right)$$

```
e6 = e5 /. HeavisideTheta[-π + t] → 0
```

$$4\,\text{Cos}[t]^2\,\text{Sin}[t]$$

Above: This is the answer for $t < \pi$.

```
e7 = e6 /. (Cos[t]^2) -> (1 - Sin[t]^2)
```

$4\,\text{Sin}[t]\,\left(1 - \text{Sin}[t]^2\right)$

```
e8 = Expand[e7]
```

$4\,\text{Sin}[t] - 4\,\text{Sin}[t]^3$

```
e9 = e8 /. (-4 Sin[t]^3) → (Sin[3 t] - 3 Sin[t])
```

$\text{Sin}[t] + \text{Sin}[3\,t]$

Above: This answer matches the text for the subinterval $0 < t < \pi$.

```
e10 = e5 /. HeavisideTheta[-π + t] → 1
```

$$4\left(\text{Cos}[t]^2\,\text{Sin}[t] - \frac{\text{Sin}[t]^3}{3}\right)$$

```
e11 = e10 /. (Sin[t]^3) → (1/4 (-Sin[3 t] + 3 Sin[t]))
```

$$4\left(\text{Cos}[t]^2\,\text{Sin}[t] + \frac{1}{12}\left(-3\,\text{Sin}[t] + \text{Sin}[3\,t]\right)\right)$$

```
e12 = Simplify[e11]
```

$$\frac{4}{3}\,\text{Sin}[3\,t]$$

Above: This version of the answer matches the text for the subinterval $t > \pi$.

23. y'' + y' - 2 y = 3 Sin[t] - Cos[t] if 0 < t < 2 π and 3 Sin[2 t] - Cos[2 t] if t > 2 π; y[0] = 1, y'[0] = 0

```
ClearAll["Global`*"]
```

```
e1 = LaplaceTransform[
  y''[t] + y'[t] - 2 y[t] == (3 Sin[t] - Cos[t]) (1 - UnitStep[t - 2 π]) +
    (3 Sin[2 t] - Cos[2 t]) (UnitStep[t - 2 π]), t, s]
```

Note that there is no shifting here; only a piecewise function definition.

$$-2\,\text{LaplaceTransform}[y[t], t, s] + s\,\text{LaplaceTransform}[y[t], t, s] + s^2\,\text{LaplaceTransform}[y[t], t, s] - y[0] - s\,y[0] - y'[0] == \frac{3}{1+s^2} - \frac{3\,e^{-2\pi s}}{1+s^2} - \frac{s}{1+s^2} + \frac{e^{-2\pi s}\,s}{1+s^2} + \frac{6\,e^{-2\pi s}}{4+s^2} - \frac{e^{-2\pi s}\,s}{4+s^2}$$

Above: Mathematica took a couple minutes’ think to come up with the transformation.

```
e2 = e1 /. {y[0] → 1, y'[0] → 0, LaplaceTransform[y[t], t, s] → bigY}
```

$$-1 - 2\,\text{bigY} - s + \text{bigY}\,s + \text{bigY}\,s^2 == \frac{3}{1+s^2} - \frac{3\,e^{-2\pi s}}{1+s^2} - \frac{s}{1+s^2} + \frac{e^{-2\pi s}\,s}{1+s^2} + \frac{6\,e^{-2\pi s}}{4+s^2} - \frac{e^{-2\pi s}\,s}{4+s^2}$$

```
e3 = Solve[e2, bigY]
```

$$\left\{\left\{\text{bigY} \to \left(e^{-2\pi s}\left(-3 + 8\,e^{2\pi s} + 3\,s - 4\,e^{2\pi s}\,s + 6\,e^{2\pi s}\,s^2 - e^{2\pi s}\,s^3 + e^{2\pi s}\,s^4\right)\right) \Big/ \left((-1+s)\left(1+s^2\right)\left(4+s^2\right)\right)\right\}\right\}$$

```
e4 = e3[[1, 1, 2]]
```

$$\left(e^{-2\pi s}\left(-3 + 8\,e^{2\pi s} + 3\,s - 4\,e^{2\pi s}\,s + 6\,e^{2\pi s}\,s^2 - e^{2\pi s}\,s^3 + e^{2\pi s}\,s^4\right)\right) \Big/ \left((-1+s)\left(1+s^2\right)\left(4+s^2\right)\right)$$

```
e5 = InverseLaplaceTransform[e4, s, t]
```

$e^t - \text{Sin}[t] - (-1 + \text{Cos}[t])\ \text{HeavisideTheta}[-2\pi + t]\ \text{Sin}[t]$

```
e6 = e5 /. HeavisideTheta[-2 π + t] → 0
```

$e^t - \text{Sin}[t]$

Above: The answer matches the text answer for the subinterval $0 < t < 2\pi$.

```
e7 = e5 /. HeavisideTheta[-2 π + t] → 1
```

$e^t - \text{Sin}[t] - (-1 + \text{Cos}[t])\ \text{Sin}[t]$

```
e8 = Simplify[e7]
```

$e^t - \text{Cos}[t]\ \text{Sin}[t]$

```
e9 = e8 /. (Cos[t] Sin[t]) → (1/2 Sin[2 t])
```

$e^t - \frac{1}{2}\text{Sin}[2\,t]$

Above: The answer matches the text answer for the subinterval $t > 2\pi$.

```
plot1 = Plot[e5, {t, 0, 8}, PlotRange → Automatic,
  PlotStyle → Thickness[0.003], ImageSize → 250]
```



Notice the gap in the above plot, which is mandated by the problem description.

25. y'' + y = t if 0 < t < 1 and 0 if t > 1; y[0] = 0, y'[0] = 0

```
ClearAll["Global`*"]
```

Starting from scratch, using the info in problem 27 as a guide. There is no shifting in the problem, only the division of the domain interval. The ‘1’ in the front of (**1 - UnitStep[*t* - 1]**) is due to zero. See the explanation in the main text on p. 220. There would be a second term, a second **UnitStep** factor, except it is multiplied by zero, the ode value above 1, so it disappears.

```
e1 = LaplaceTransform[y''[t] + y[t] == t (1 - UnitStep[t - 1]), t, s]
```

$$\text{LaplaceTransform}[y[t], t, s] + s^2\,\text{LaplaceTransform}[y[t], t, s] - s\,y[0] - y'[0] == \frac{1}{s^2} - \frac{e^{-s}\,(1+s)}{s^2}$$

```
e2 = e1 /. {y[0] → 0, y'[0] → 0, LaplaceTransform[y[t], t, s] → bigY}
```

$$\text{bigY} + \text{bigY}\,s^2 == \frac{1}{s^2} - \frac{e^{-s}\,(1+s)}{s^2}$$

```
e3 = Solve[e2, bigY]
```

$$\left\{\left\{\text{bigY} \to \frac{e^{-s}\,(-1 + e^s - s)}{s^2\,(1+s^2)}\right\}\right\}$$

```
e4 = e3[[1, 1, 2]]
```

$$\frac{e^{-s}\,(-1 + e^s - s)}{s^2\,(1+s^2)}$$

```
e5 = InverseLaplaceTransform[e4, s, t]
t - HeavisideTheta[-1 + t] (t - Cos[1 - t] + Sin[1 - t]) - Sin[t]
```

```
e6 = e5 /. HeavisideTheta[-1 + t] → 0
```

```
t - Sin[t]
```

Above answer is for $t < 1$, and it matches the text answer for this subinterval.

```
e7 = e5 /. HeavisideTheta[-1 + t] → 1
```

```
Cos[1 - t] - Sin[1 - t] - Sin[t]
```

Above for $t > 1$. Since Cos is even function and Sin is odd, the answer matches the text answer for this subinterval.

```
y[t_] = t - HeavisideTheta[-1 + t] (t - Cos[1 - t] + Sin[1 - t]) - Sin[t]
t - HeavisideTheta[-1 + t] (t - Cos[1 - t] + Sin[1 - t]) - Sin[t]
```

```
y[0]
0
```

```
y'[0]
0
```

```
e8 = y'[t]
1 - Cos[t] - HeavisideTheta[-1 + t] (1 - Cos[1 - t] - Sin[1 - t]) -
 DiracDelta[-1 + t] (t - Cos[1 - t] + Sin[1 - t])
```

```
plot1 = Plot[e5, {t, 0, 3}, PlotRange → Automatic,
   PlotStyle → Thickness[0.003], ImageSize → 250];
plot2 = Plot[e8, {t, 0, 3}, PlotRange → Automatic,
   PlotStyle → {Red, Thickness[0.003]}, ImageSize → 250];
```

```
Show[plot1, plot2]
```



An interesting plot. The initial condition points seem to work after all.

27. Shifted data. $y'' + 4y = 8t^2$ if $0 < t < 5$ and 0 if $t > 5$;
$y[1] = 1 + \text{Cos}2$, $y'[1] = 4 - 2\,\text{Sin}[2]$

```
ClearAll["Global`*"]
```

The initial conditions are given at $t = 1$, but the transform calculates them for $t = 0$; it will be necessary to shift, with $\tilde{t} = t - 1$. Also, the unit step function comes into play, with a step of $t - 5$, or $\tilde{t} - 4$. Together, this changes the problem equation to:
$\tilde{y}''[1+\tilde{t}] + 4\,\tilde{y}[1+\tilde{t}] == 8\,(1+\tilde{t})^2\,(1 - u[\tilde{t} - 4])$. (The changed version supplied by the answer in the text.)

In the version below, down to e7, $\tilde{t}$ is intended, but the symbol used is just t.

```
e1 = LaplaceTransform[y''[t] + 4 y[t] == 8 (1 + t)^2 (1 - UnitStep[t - 4]), t, s]
```

$$4\,\text{LaplaceTransform}[y[t],t,s] + s^2\,\text{LaplaceTransform}[y[t],t,s] - s\,y[0] - y'[0] == 8\left(\frac{2}{s^3}+\frac{2}{s^2}+\frac{1}{s}-\frac{e^{-4s}}{s}-\frac{2e^{-4s}(1+4s)}{s^2}-\frac{2e^{-4s}\left(1+4s+8s^2\right)}{s^3}\right)$$

Above: This is the best transcription I can manage of the equation as shown by the text.

```
e2 = e1 /. {y[0] → 1 + Cos[2],
   y'[0] → 4 - 2 Sin[2], LaplaceTransform[y[t], t, s] → bigY}
```

$$-4 + 4\,\text{bigY} + \text{bigY}\,s^2 - s\,(1+\text{Cos}[2]) + 2\,\text{Sin}[2] == 8\left(\frac{2}{s^3}+\frac{2}{s^2}+\frac{1}{s}-\frac{e^{-4s}}{s}-\frac{2e^{-4s}(1+4s)}{s^2}-\frac{2e^{-4s}\left(1+4s+8s^2\right)}{s^3}\right)$$

```
e3 = Solve[e2, bigY]
```

$$\left\{\left\{\text{bigY}\to\frac{1}{s^3\left(4+s^2\right)}e^{-4s}\left(-16+16e^{4s}-80s+16e^{4s}s-200s^2+8e^{4s}s^2+4e^{4s}s^3+e^{4s}s^4+e^{4s}s^4\,\text{Cos}[2]-2e^{4s}s^3\,\text{Sin}[2]\right)\right\}\right\}$$

```
e4 = e3[[1, 1, 2]]
```

$$\frac{1}{s^3\left(4+s^2\right)}e^{-4s}\left(-16+16e^{4s}-80s+16e^{4s}s-200s^2+8e^{4s}s^2+4e^{4s}s^3+e^{4s}s^4+e^{4s}s^4\,\text{Cos}[2]-2e^{4s}s^3\,\text{Sin}[2]\right)$$

```
e5 = InverseLaplaceTransform[e4, s, t]
```

$$1+4t+2t^2+\text{Cos}[2(1+t)]-\text{HeavisideTheta}[-4+t]\left(1+4t+2t^2-49\,\text{Cos}[8-2t]+10\,\text{Sin}[8-2t]\right)$$

```
e6 = Simplify[e5]
```

$$1+4t+2t^2+\text{Cos}[2(1+t)]-\text{HeavisideTheta}[-4+t]\left(1+4t+2t^2-49\,\text{Cos}[8-2t]+10\,\text{Sin}[8-2t]\right)$$

```
e7 = e6 /. t → t - 1
```

$$1+4(-1+t)+2(-1+t)^2+\text{Cos}[2t]-\text{HeavisideTheta}[-5+t]\left(1+4(-1+t)+2(-1+t)^2-49\,\text{Cos}[8-2(-1+t)]+10\,\text{Sin}[8-2(-1+t)]\right)$$

Above: This restores the meaning of t. It is no longer $\tilde{t}$, but itself, regular t.

```
e8 = Expand[e7]
```

$$-1+2t^2+\text{Cos}[2t]+\text{HeavisideTheta}[-5+t]-2t^2\,\text{HeavisideTheta}[-5+t]+49\,\text{Cos}[8-2(-1+t)]\,\text{HeavisideTheta}[-5+t]-10\,\text{HeavisideTheta}[-5+t]\,\text{Sin}[8-2(-1+t)]$$

The above matches the text answer, if all the factors incorporating the **HeavisideTheta** function are evaluated according to the function definition, i.e., function equals 0 for argument less than zero.

```
e15 = e8 /. HeavisideTheta[-5 + t] → 0
```

$-1 + 2\,t^2 + \text{Cos}[2\,t]$

The **HeavisideTheta** only acts as a switch to turn the adjoining factors ‘on’. For argument greater than zero, the Heaviside function returns 1. Thus the above answer is for the subinterval $0 < t < 5$.

```
e9 = e8 /. HeavisideTheta[-5 + t] → 1
49 Cos[8 - 2 (-1 + t)] + Cos[2 t] - 10 Sin[8 - 2 (-1 + t)]

e10 = Simplify[e9]
49 Cos[2 (-5 + t)] + Cos[2 t] - 10 Sin[10 - 2 t]

e11 = -Sin[x] == Sin[-x]
True

e14 = Cos[x] == Cos[-x]
True

e12 = e10 /. {2 (-5 + t) → -10 + 2 t}
```

49 Cos[10 - 2 t] + Cos[2 t] - 10 Sin[10 - 2 t]

The above answer is equivalent to the text answer. The only differences are the signs of the arguments inside Sin and Cos, which, because of odd Sin and even Cos, agree with the text intent. For some reason I cannot use substitution commands to make Mathematica change these signs. The subinterval which this answer applies to is $t > 5$.

38 - 40 RLC-circuit.
Using the Laplace transform and showing the details, find the current i(t) in the circuit in the figure below, assuming zero initial current and charge and:

39. R = 2 Ω, L = 1 H, C = 0.5 F, v (t) = 1 kV if 0 < t < 2 and 0 if t > 2



RLC–circuit (problems 38 – 40)

In[60]:=
```
ClearAll["Global`*"]
```

First, setting up the electrical state space model just as if the domain were not piecewise.

In[61]:=
```
eqns = {eL q''[t] + aR q'[t] + 1/cC q[t] == Vee[t]};
```

In[62]:=
```
m1 = StateSpaceModel[eqns,
  {{q[t], 0}, {q'[t], 0}}, {{Vee[t], 0}}, {q'[t]}, t]
```

Out[62]=

$$\left(\begin{array}{cc|c} 0 & 1 & 0 \\ -\frac{1}{cC\ eL} & -\frac{aR}{eL} & \frac{1}{eL} \\ \hline 0 & 1 & 0 \end{array}\right)^{S}$$

And putting in the capacitance, inductance, and resistance from the problem description.

In[63]:=
```
mw = m1 /. {cC → 0.5, eL → 1, aR → 2}
```

Out[63]=

$$\left(\begin{array}{cc|c} 0 & 1 & 0 \\ -2. & -2 & 1 \\ \hline 0 & 1 & 0 \end{array}\right)^{S}$$

And getting an output response for the model. It’s very pleasing to see how well and effortlessly Mathematica accommodates a **Piecewise** function for the voltage input. Perhaps Laplace was used under the hood, but all that really matters is that the output is as expected.

In[64]:=
```
outpw = OutputResponse[{mw},
  {Piecewise[{{1000, 0 < t < 2}, {0, 2 < t}}]}, {t, 0, 5}]
```

Out[64]=
```
{InterpolatingFunction[ Domain {{0., 5.}} Output scalar ][t]}
```

The plot agrees with that shown in the s.m., as will be seen in a minute.

In[126]:=
```
p6 = Plot[{outpw}, {t, 0, 5}, ImageSize → 400, AspectRatio → 0.5,
   PlotRange → All, PlotStyle → {Red, Thickness[0.01]}, GridLines → All];
```

The disadvantage to the **InterpolatingFunction** object is that it isn't a real equation. So I need to press on a little, first by making a table of values. The two fabricated polys (yellow) are not exact, but can obviously be used in many instances, and could be sharpened to almost any desired margin of error.

```
In[66]:= gv = Table[{t, outpw[t]}, {t, 0, 2, 0.05}];
```

Marco, at Wolfram Community, gave me the key to cleaning up the output of the above table and making it a suitable list for plotting.

```
In[67]:= gv2 = {#[[1]], #[[2, 0, 1]]} & /@ gv;

In[68]:= ListPlot[gv2];
```

The fabricated function below will fit pretty well to the interpolated function in the range 0 - 2 with only a 4th power poly.

In[69]:= duke = Fit$\left[gv2, \left\{x^4, x^3, x^2, x, x^0\right\}, x\right]$

Out[69]= $0.0950882 + 999.827\,x - 1007.86\,x^2 + 365.813\,x^3 - 48.2548\,x^4$

```
In[80]:= p1 = Plot[duke,  {x, 0, 2}, PlotStyle → {White, Thickness[0.002]}];

In[99]:= gv3 = Table[{t, outpw[t]}, {t, 2, 5, 0.05}];

In[100]:= gv4 = {#[[1]], #[[2, 0, 1]]} & /@ gv3;

In[101]:= ListPlot[gv4];
```

Because of the discontinuity at t=2, there need to be two separate polys. The fit for the fabricated function in the interval 2 - 5 needs a 5th power poly to look good.

In[102]:= duke2 = Fit$\left[gv4, \left\{x^5, x^4, x^3, x^2, x, x^0\right\}, x\right]$

Out[102]= $11\,601.8 - 13\,304.7\,x + 5686.9\,x^2 - 1167.05\,x^3 + 116.666\,x^4 - 4.57576\,x^5$

```
In[103]:= p2 = Plot[duke2, {x, 2, 5}, PlotStyle → {White, Thickness[0.002]}];
```

The fabricated poly fits well over the interpolating function.

```
In[104]:= Show[p6, p1, p2,  PlotRange → All]
```

Out[104]=



The following plot shows the interpolating function (same as plot p6) along with the text answer (p7). It can be seen from the plots that the interpolated solution follows the exact (Laplace) solution closely.

In[123]:=
```
p7 = Plot[Piecewise[{{1000 e^-t Sin[t], 0 < t < 2}, {1000 e^-t Sin[t] -
      1000 UnitStep[t - 2] e^-(t-2) Sin[t - 2], 2 < t}}, {t, 0, 5}], {t, 0, 5},
   PlotStyle → {RGBColor[0.4, 0.8, 0.3], Thickness[0.006]}];
```

In[117]:=
```
p8 = Plot[{outpw}, {t, 0, 5},
   ImageSize → 400, AspectRatio → 0.5, PlotRange → All,
   PlotStyle → {White, Thickness[0.002]}, GridLines → All];
```

In[124]:=
```
Show[p7, p8]
```

Out[124]=