Note: In this problem set, expressions in green cells match corresponding expressions in the text answers.

```
Clear["Global`*"]
```

1 - 9 Integrals involving sine and cosine
Evaluate the following integrals.

1. $\int_0^{\pi} \frac{2}{k - \text{Cos}[\theta]} \, d\theta$

```
Clear["Global`*"]
```

First I will search for singularities.

```
Reduce[k - Cos[θ] == 0, {k, θ}]
```

```
C[1] ∈ Integers && (θ == -ArcCos[k] + 2 π C[1] || θ == ArcCos[k] + 2 π C[1])
```

The cells both above and below show the pattern for multiples of ArcCos[k], and allow me to use simply ArcCos[k]+2 π as the root. However, looking at the interval of evaluation for the integral, only C[1]=0 will be available. For real k, this root will be on the x axis, constituting a simple pole, and that permits use of theorem 1 on p. 731.

The above mentioned theorem 1 sets forth the answer to the integral as

$$\pi\, i\, \text{Residue}\left[\frac{2}{k - \text{Cos}[\theta]}, \{\theta, \text{ArcCos}[k]\}\right]$$

$$\frac{2\, i\, \pi}{\sqrt{1 - k^2}}$$

Or,

$$\frac{2\, i\, \pi}{\sqrt{1 - k^2}} == \frac{2\, i\, \pi}{\sqrt{k^2 - 1}\ i} == \frac{2\ \pi}{\sqrt{k^2 - 1}}$$

3. $\int_0^{2\pi} \frac{1 + \text{Sin}[\theta]}{3 + \text{Cos}[\theta]} \, d\theta$

```
Clear["Global`*"]
```

$$\text{Integrate}\left[\frac{1 + \text{Sin}[\theta]}{3 + \text{Cos}[\theta]}, \{\theta, 0, 2\pi\}\right]$$

$$\frac{\pi}{\sqrt{2}}$$

I tried this a long way first, involving residue, but did not get the right answer. Just by pushing the integrate button, the right answer pops out.

5. $\int_0^{2\pi} \frac{\text{Cos}[\theta]^2}{5 - 4\,\text{Cos}[\theta]} \, d\theta$

```
Clear["Global`*"]
```

$$\text{Integrate}\left[\frac{\text{Cos}[\theta]^2}{5 - 4\,\text{Cos}[\theta]}, \{\theta, 0, 2\pi\}\right]$$

$$\frac{5\pi}{12}$$

Another one matches the text answer without any preparation or application.

7. $\int_0^{2\pi} \frac{a}{a - \text{Sin}[\theta]} \, d\theta$

```
Clear["Global`*"]
```

$$\text{Integrate}\left[\frac{a}{a - \text{Sin}[\theta]}, \{\theta, 0, 2\pi\}\right]$$

$$2\sqrt{\frac{a^2}{-1 + a^2}}\,\pi == \frac{2\sqrt{a^2}}{\sqrt{1 - a^2}}\,\pi == \frac{-2\,a\,\pi}{\sqrt{a^2 - 1}}$$

Mathematica gives an answer with reversed sign, as compared to the text answer. (Mathematica took a long think on this.) Since this is the only problem in this section where there is disagreement in answers, I will attempt to work it out the long way.

$$\int_0^{2\pi} \frac{a}{a - \text{Sin}[\theta]} \, d\theta == a \int_0^{2\pi} \frac{1}{a - \text{Sin}[\theta]} \, d\theta$$

True

Numbered line (2) on p. 726 has a way to modify the appearance of the denominator,

$$a - \text{Sin}[\theta] == a - \frac{1}{2\,i}\left(z - \frac{1}{z}\right)$$

and in the text below numbered line (2),

$$d\theta = \frac{dz}{i\,z}$$

Implying that

$$\int_0^{2\pi} \frac{1}{a - \text{Sin}[\theta]} \, d\theta == \oint_C \frac{dz}{i\,z\left(a - \frac{1}{2\,i}\left(z - \frac{1}{z}\right)\right)}$$

where C is the unit circle.

Operating on the denominator above,

$$i\, z \left(a - \frac{1}{2\, i} \left(z - \frac{1}{z}\right)\right) == i\, z\, a - \frac{i\, z}{2\, i}\, z + \frac{1}{2\, i}\, \frac{i\, z}{z} ==$$

$$i\, z\, a - \frac{z^2}{2} + \frac{1}{2} == -\frac{1}{2}\left(z^2 - 2\, a\, i\, z - 1\right)$$

making the last integral equal to

$$-2 \oint_C \frac{dz}{z^2 - 2\, a\, i\, z - 1}$$

The denominator is a quadratic equation, with a=1, b=-2 a i , and c=-1. Solving,

```
Solve[z^2 - 2 a ⅈ z - 1 == 0, z]
```

$$\left\{\left\{z \to i\, a - \sqrt{1 - a^2}\right\}, \left\{z \to i\, a + \sqrt{1 - a^2}\right\}\right\}$$

The roots will be simple poles of the problem function. A plot would be appropriate at this point.

```
p1 = ParametricPlot[{1 Cos[t], 1 Sin[t]}, {t, -π, π},
   ImageSize → 500, AxesLabel → {"Re", "Im"}, PlotRange → {-1.5, 1.5},
   PlotStyle → {Thickness[0.004]}, GridLines -> Automatic];

p1p = ListPlot[Table[{Re[ⅈ a - Sqrt[1 - a^2]], Im[ⅈ a - Sqrt[1 - a^2]]}, {a, -5, 5}],
   PlotStyle → {Red}];

p2p = ListPlot[Table[{Re[ⅈ a + Sqrt[1 - a^2]], Im[ⅈ a + Sqrt[1 - a^2]]}, {a, -5, 5}],
   PlotStyle → {Blue}];
```

```
Show[p1, p1p, p2p]
```



There appear to be 8 points inside the unit circle. For the first table below (red points), these are n=2,3,4,5.

```
TableForm[Table[{a, N[Re[i a - √(1 - a²)]], N[Im[i a - √(1 - a²)]],
   Re[i a - √(1 - a²)], Im[i a - √(1 - a²)]}, {a, -5, 5}]]
```

-5	0.	-9.89898	0	$-5-2\sqrt{6}$
-4	0.	-7.87298	0	$-4-\sqrt{15}$
-3	0.	-5.82843	0	$-3-2\sqrt{2}$
-2	0.	-3.73205	0	$-2-\sqrt{3}$
-1	0.	-1.	0	-1
0	-1.	0.	-1	0
1	0.	1.	0	1
2	0.	0.267949	0	$2-\sqrt{3}$
3	0.	0.171573	0	$3-2\sqrt{2}$
4	0.	0.127017	0	$4-\sqrt{15}$
5	0.	0.101021	0	$5-2\sqrt{6}$

A series like the below may not be too smart, but it is possible.

```
Series[a/(a - Sin[θ]), {a, 5 - 2 √6, 4}]
```

Which I take to mean that I could examine residues in terms of *a* if I wanted to.

```
Residue[a/(a - Sin[θ]), {a, 5 - 2 √6}]
```

0

```
Residue[a/(a - Sin[θ]), {a, 4 - 2 √6}]
```

0

```
Residue[a/(a - Sin[θ]), {a, 3 - 2 √2}]
```

0

```
Residue[a/(a - Sin[θ]), {a, 2 - √3}]
```

0

The cells above are probably an abuse of the Residue function. In any case, all are zero. Next are the blue points, n=-5,-4,-3,-2.

```
TableForm[Table[{a, N[Re[ⅈ a + √(1 - a²)]], N[Im[ⅈ a + √(1 - a²)]],
   Re[ⅈ a + √(1 - a²)], Im[ⅈ a + √(1 - a²)]}, {a, -5, 5}]]
```

-5	0.	-0.101021	0	$-5+2\sqrt{6}$
-4	0.	-0.127017	0	$-4+\sqrt{15}$
-3	0.	-0.171573	0	$-3+2\sqrt{2}$
-2	0.	-0.267949	0	$-2+\sqrt{3}$
-1	0.	-1.	0	-1
0	1.	0.	1	0
1	0.	1.	0	1
2	0.	3.73205	0	$2+\sqrt{3}$
3	0.	5.82843	0	$3+2\sqrt{2}$
4	0.	7.87298	0	$4+\sqrt{15}$
5	0.	9.89898	0	$5+2\sqrt{6}$

```
Residue[a/(a - Sin[θ]), {a, -5 + 2 √6}]
```

0

```
Residue[a/(a - Sin[θ]), {a, -4 + √15}]
```

0

```
Residue[a/(a - Sin[θ]), {a, -3 + 2 √2}]
```

0

```
Residue[a/(a - Sin[θ]), {a, -2 + √3}]
```

0

The residues of all the blue points also equal zero. What has been accomplished, at least, is to find eight values of ***a*** which, when inserted into the root formula, produce roots inside the unit circle. So returning to the problem function, I can ask for a table of *a* values, like so

```
TableForm[Table[{a, Integrate[a/(a - Sin[θ]), {θ, 0, 2 π}]},
  {a, {-5, -4, -3, -2, 2, 3, 4, 5}}]]
```

-5	$\frac{5\pi}{\sqrt{6}}$
-4	$\frac{8\pi}{\sqrt{15}}$
-3	$\frac{3\pi}{\sqrt{2}}$
-2	$\frac{4\pi}{\sqrt{3}}$
2	$\frac{4\pi}{\sqrt{3}}$
3	$\frac{3\pi}{\sqrt{2}}$
4	$\frac{8\pi}{\sqrt{15}}$
5	$\frac{5\pi}{\sqrt{6}}$

All of the resulting values conform to the text answer, $\frac{2\,a\,\pi}{\sqrt{a^2-1}}$, and these test cases carry a positive sign. Maybe this suggests that *Mathematica*'s answer was in error. In any case it is puzzling that no minus sign is generated when evaluating integer tokens, but a minus sign is generated when evaluating symbolic tokens. What if the integral is presented in a different way?

```
Integrate[a/(a - Sin[θ]), {θ, 0, 2 π}, Assumptions → a ∈ Integers]
```

```
ConditionalExpression[(2 a π Sign[a])/√(-1 + a²),
 Im[ArcSin[a]] ≠ 0 || Re[ArcSin[a]] > π || π + Re[ArcSin[a]] < 0]
```

The numerator of the conditional expression is structured to remain positive. This could be used as a definitive result, but I'll continue, next with a try at splitting up the integral into two pieces.

```
Integrate[a/(a - Sin[θ]), {θ, 0, 2 π}, Assumptions → -5 ≤ a ≤ -2]
```

$$-\frac{2 a \pi}{\sqrt{-1 + a^2}}$$

The above cell agrees with the test results for specific negative values of a.

```
Integrate[a/(a - Sin[θ]), {θ, 0, 2 π}, Assumptions → 2 ≤ a ≤ 5]
```

$$\frac{2 a \pi}{\sqrt{-1 + a^2}}$$

The above cell agrees with the test results for specific positive values of *a*. At this point it looks like Mathematica may have a better adapted solution, whereas the text answer may fail on blue points. (Checking again, I cannot see that negative values of *a* are prohibited.) The case may be better defined, but it is still a case of disagreement.

9. $\int_0^{2\pi} \frac{\cos[\theta]}{13 - 12 \cos[2\theta]} \, d\theta$

```
Clear["Global`*"]
```

```
Integrate[Cos[θ]/(13 - 12 Cos[2 θ]), {θ, 0, 2 π}]
```

0

10 - 22 Improper integrals: Infinite interval of integration
Evaluate the following integrals.

11. $\int_{-\infty}^{\infty} \frac{1}{(1 + x^2)^2} \, dx$

```
Clear["Global`*"]
```

```
Integrate[1/(1 + x^2)^2, {x, -∞, ∞}]
```

$$\frac{\pi}{2}$$

13. $\int_{-\infty}^{\infty} \frac{x}{\left(1+x^2\right)\left(x^2+4\right)} \, dx$

```
Clear["Global`*"]
```

```
Integrate[x/((1 + x^2) (x^2 + 4)), {x, -∞, ∞}]
```

0

15. $\int_{-\infty}^{\infty} \frac{x^2}{x^6+1} \, dx$

```
Clear["Global`*"]
```

```
Integrate[x^2/(x^6 + 1), {x, -∞, ∞}]
```

$\frac{\pi}{3}$

17. $\int_{-\infty}^{\infty} \frac{\text{Sin}[3\,x]}{x^4+1} \, dx$

```
Clear["Global`*"]
```

$\int_{-\infty}^{\infty} \frac{\text{Sin}[3\,x]}{x^4+1} \, dx$

0

19. $\int_{-\infty}^{\infty} \frac{1}{x^4-1} \, dx$

```
Clear["Global`*"]
```

```
Integrate[1/(x^4 - 1), {x, -∞, ∞}, PrincipalValue → True]
```

$-\frac{\pi}{2}$

In this problem, I got a non-converge warning until I added the request for principal value.

21. $\int_{-\infty}^{\infty} \frac{\text{Sin}[x]}{(x-1)\left(x^2+4\right)} \, dx$

```
Clear["Global`*"]
```

```
Integrate[Sin[x]/((x - 1) (x^2 + 4)), {x, -∞, ∞}, PrincipalValue → True]
```

$$\frac{1}{5}\pi\left(-\frac{1}{e^2}+\text{Cos}[1]\right)$$

In this problem, I got a non-converge warning until I added the request for principal value.

23 - 26 Improper integrals: Poles on the real axis
Find the Cauchy principal value.

23. $\int_{-\infty}^{\infty}\frac{1}{x^4-1}\,dx$

```
Clear["Global`*"]
```

```
Integrate[1/(x^4 - 1), {x, -∞, ∞}, PrincipalValue → True]
```

$$-\frac{\pi}{2}$$

25. $\int_{-\infty}^{\infty}\frac{x+5}{x^3-x}\,dx$

```
Clear["Global`*"]
```

```
Integrate[(x + 5)/(x^3 - x), {x, -∞, ∞}, PrincipalValue → True]
```

0

All of the green cells in this section contain expressions which agree with the text answers.